



Technical Note 03-01

レーダーグラフ

By Olivier Deschanel, 4D S.A.

Technical Note 03-01

(原題: HTTP Kiviat Graphs)

概要

Kiviat Graph、通称・レーダーグラフでは、放射状の軸を用いて、カテゴリ別のデータを表現します。レーダーグラフは、図形の形状からある程度データの特徴を知ることができ、図形を重ねて出力すれば、データの比較が容易です。円に近いグラフほどバランスが良く、いびつな形は極端なデータを意味します。

4D Chart

4D Chart のドキュメントを見ても、レーダーというチャートタイプは存在しません。また、プログラムコードなしでは同一のエリアに複数のチャートを重ねて出力できないという制限もあります。とはいえ、線、円、多角形を描画することはできるので、4D のコマンドをうまく使えば、レーダーグラフを作成することができます。

以降の説明では、600×600 ピクセルの 4D Chart エリア zChart とボタンが存在するものと仮定し、そのボタンのオブジェクトメソッドでエリアの制御をしています。

(サンプルデータベースの GK_demo フォーム)

```
CT DO COMMAND(zChart;1001)
```

エリアを初期化します。1001 はファイルメニュー項目の「新規」です。

```
zCenter_x:=300  
zCenter_y:=300  
zRadius:=300
```

グラフの中心点と半径を定義します。

```
ARRAY REAL(_values;6)  
_values{1}:=3  
_values{2}:=2.5  
_values{3}:=4  
_values{4}:=1  
_values{5}:=4  
_values{6}:=3.5
```

グラフに使用されるデータを実数の配列に代入します。

```
ARRAY STRING(20;_legend_series;6)
_legend_series{1}:="Resolution"
_legend_series{2}:="Color"
_legend_series{3}:="Sensitivity"
_legend_series{4}:="Flash"
_legend_series{5}:="Autofocus"
_legend_series{6}:="Sensitivity"
```

盆例に使用される名義を設定します。

```
zNbValues:=Size of array(_values)
$max:=0
For ($i;1;zNbValues;1)
  If ($max<_values{$i})
    $max:=$_values{$i}
  End if
End for
$max:=Int($max)+1
```

表示する最大値を算出します。

これで実際の描画に取りかかることができます。すべてが重ねられて表示されるように、正しい順序で出力しなくてはなりません。

- 1- クモの巣のような背景
- 2- データの多角形
- 3- 軸
- 4- 目盛り
- 5- 軸の名前

軸の角度は、軸の数によって決定されます。これら一連の角度は、何度か使用するので、配列に保存しておきます。三角関数では、反時計回りが標準ですが、円グラフやレーダーグラフでは、時計回りが読み書きの標準なので注意が必要です。また、グラフは 12 時を起点とするのに対し、0 ラジアンは 3 時なので、オフセットを使用してその点を調整します。値をラジアンに変換するには Degree が便利です。

```
ARRAY REAL($_angle;zNbValues)
$direction:=-1$begin_angle:=90*Degree
For ($point;1;zNbValues)
  $_angle{$point}:=$begin_angle+(360*Degree/zNbValues*($point-1)*$direction)
End for
```

多角形の描画には x 軸と y 軸の配列を使用します。多角形を閉じるために最初の座標に戻るため、配列の要素数は角の数+1 となります。

```
ARRAY REAL($_x;zNbValues+1)
ARRAY REAL($_y;zNbValues+1)
```

今回は値の範囲が 1-5 だと分かっているので、目盛りの単位は 1 とします。異なる条件では、この値を修正する必要があります。

\$Step:=1

クモの巣のような多角形は形が同じなので、倍率をかけて描画できます。グラフにおける y 軸の数は、コンピュータの座標とは逆であり、始点も異なるので次のオフセットを使用します。

Xscreen = Xcenter + Xtrigo

Yscreen = Ycenter - Ytrigo

```
▼ For ($line;1;$max;$step)
  $line_radius:=zRayon*$line/$max
  ▼ For($point;1;zNbValues)
    $x{$point}:=zCentre_x+(Cos($_angle{$point}))*$line_radius
    $y{$point}:=zCentre_y-(Sin($_angle{$point}))*$line_radius
  End for
  $x{zNbValues+1}:=$x{1}
  $y{zNbValues+1}:=$y{1}
  $poly:=CT Array to polygon (zChart;$x;$y)
  CT SET FILL ATTRIBUTES (zChart;$poly;1;-1)
End for
```

グラフの多角形も同様ですが、座標の値は個別に計算する点が異なります。この値は軸の最大長と値の最大値に対する比率で求められます。

```
▼ For ($point;1;zNbValues)
  $x{$point}:=zCentre_x+(Cos($_angle{$point}))*zRayon*_values{$point}/$max
  $y{$point}:=zCentre_y-(Sin($_angle{$point}))*zRayon*_values{$point}/$max
End for
$x{zNbValues+1}:=$x{1}
$y{zNbValues+1}:=$y{1}
$polygon:=CT Array to polygon (zChart;$x;$y)
CT SET FILL ATTRIBUTES (zChart;$polygon;3;CT Index to color (4))
```

放射状の軸は配列の角度を使用して中心点から描画します。

```
▼ For ($point;1;zNbValues)
  $line:=CT Draw line (zChart;zCentre_x;zCentre_y;zCentre_x+
    (Cos($_angle{$point}))*zRayon;zCentre_x-(Sin($_angle{$point}))*zRayon))
End for
```

主要な軸(通常は垂直)の左隣に目盛りの値をテキストで表示します。

```
▼ For ($line;1;$max;$step)
  $point:=1
  $line_radius:=zRayon*$line/$max
  $x:=zCentre_x+(Cos($_angle{$point}))*$line_radius
  $y:=zCentre_y-(Sin($_angle{$point}))*$line_radius
  $text:=CT Draw text (zChart;$x-20;$y+5;$x-2;$y+15;String($line))
  CT SET TEXT ATTRIBUTES (zChart;$text;CT Font number ("Geneva");9;Bold;CT Index to color (16);2)
End for
```

仕上げてそれぞれの軸に凡例を表示します。位置計算の If 文に工夫があります。

```

▼ For ($point;1;zNbValues)
  $x:=zCentre_x+(Cos($_angle{$point})*zRayon)
  $y:=zCentre_x-(Sin($_angle{$point})*zRayon)
  ▼ If ($x>=zCentre_x)
    $x1:=$x+10
    $x2:=$x+100
    $justif:=0
  ▼ Else
    $x1:=$x-10
    $x2:=$x-100
    $justif:=2
  End if
  ▼ If ($y>=zCentre_y)
    $y1:=$y+5
    $y2:=$y+15
  ▼ Else
    $y1:=$y-5
    $y2:=$y-15
  End if
  $text:=CT Draw text(zChart;$x1;$y1;$x2;$y2;_legend_series{$point})
  CT SET TEXT ATTRIBUTES (zChart;$text;CT Font number ("Geneva");9;Plain;CT Index to color (16);$justif)
End for

```

